
The Big Picture

We are interested in causal effects of a treatment (cause) X on an outcome Y .

We are not only interested in the average causal effect, but also in conditional causal
effects (e.g., effects on the treated, effects for patients with a high neediness, effects for
females,...).

We compute causal effects based on an unbiased regression E(Y |X ,K ,Z), where K are
categorical covariates and Z are continuous covariates.

The covariates must be selected in such a way, that unbiasedness of E(Y |X ,K ,Z) holds.
While unbiased itself is not testable, some of the (stronger) causality conditions
implying unbiasedness can be tested empirically.

Based on the unbiased regression E(Y |X ,K ,Z), we compute average and conditional
causal effects using a multigroup SEM approach.

Introduction

We present a multigroup structural equation modeling approach that allows for the
identification of average and conditional effects of a discrete treatment variable X on a
continuous outcome variable Y , taking into account categorical covariates K and continuous
covariates Z .

The treatment variable X can take on values 0,1, . . . ,T and the regression E(Y |X ,K ,Z) can
always be parameterized as follows:

E(Y |X ,K ,Z) = g0(K ,Z)+g1(K ,Z) · IX=1 + . . .+gT (K ,Z) · IX=T ,

where g0(K ,Z), . . . ,gT (K ,Z) are any functions of K ,Z and IX=t is an indicator variable for
X = t. We call the X = 0 the ’control group’ and all other treatment groups X = 1, . . . ,T are
compared to it.

Average and Conditional Effects

Average and conditional effects:

The values of the gt (K ,Z)-functions are the conditional treatment effects of treatment X = t
vs. treatment X = 0 given values of K and Z .

We can consider expectations and conditional expectations of the gt (K ,Z)-functions, i.e.,
average and conditional effects:

E[gt (K ,Z)] Average effect (AVE)
gt (k,z) Conditional effect given a value z of Z and a value k of K
E[gt (K ,Z) |X = t∗] Effect given X = t∗ (Effect on the treated)
E[gt (K ,Z) |X = 0] Effect on the non-treated
E[gt (K ,Z) |K = k] Effect given a value K = k
E[gt (K ,Z) |Z = z] Effect given a value Z = z
E[gt (K ,Z) |V = v] Effect given a value V = v, where V = f (X ,K ,Z)

Axel Mayer (2014) Causal Effects in Mplus 21.07.14 5 / 40



Multigroup Structural Equation Models

We will use multigroup structural equation models to estimate these average and conditional
effects. The approach uses as many groups as there are combinations of treatment groups
and levels of categorical covariates. Then, the regressions of the dependent variable on the
continuous covariates are estimated in each cell and adjusted means as well as average and
conditional causal effects are calculated.

The multigroup structural equation modeling approach allows for

1 latent covariates and latent outcome variables,

2 (higher order) interactions between the treatment variable and categorical and
continuous covariates,

3 stochastic and fixed categorical and continuous covariates, and

4 heterogeneous residual variances across groups.

Multigroup Structural Equation Models

The multigroup structural equation model consists of a group-specific measurement model
and a group-specific structural model:(

Y
Z

)
=νg +Λg η + ε Measurement model for group g

η=αg +Bg η+ζw Structural model for group g

The multigroup SEM implies a certain structure for means and (co-)variances of observed
variables and the parameters of the model are estimated using ML (or other estimators).

In order to compute average effects, we also need a model for the group sizes
(KNOWNCLASS option in Mplus):

P(X = m) = exp(λm)

1+
n−1∑
i=0

exp(λi)

0 ≤ m ≤ n−1

P(X = n) = 1−
n−1∑
m=0

P(X = m)

Example 1

Axel Mayer (2014) Causal Effects in Mplus 21.07.14 8 / 40



Example 1

Generalized ANCOVA with two treatment conditions, linear g-functions, and a single
continuous covariate Z :

E(Y |X ,Z) = g0(Z)+g1(Z) ·X

= (γ00 +γ01 ·Z)+ (γ10 +γ11 ·Z) ·X .

Some average and conditional effects:

E[g1(Z)] =γ10 +γ11 ·E(Z) Average effect (AVE)
g1(z) =γ10 +γ11 ·z Conditional effect give a value z of Z
E[g1(Z) |X = 1] =γ10 +γ11 ·E(Z |X = 1) Effect on the treated
E[g1(Z) |X = 0] =γ10 +γ11 ·E(Z |X = 0) Effect on the non-treated

Example 1

Multigroup SEM:(
Y
Z

)
=

(
β00
µ01

)
+

(
β01

0

)
Z +

(
ζ1
ζ2

)
Structural model X = 0(

Y
Z

)
=

(
β10
µ11

)
+

(
β11

0

)
Z +

(
ζ1
ζ2

)
Structural model X = 1

P(X = 0) = exp(λ0)

1+exp(λ0)
Group size X = 0

P(X = 1) = 1−P(X = 0) Group size X = 1

(Measurement model is omitted, because there are no latent variables.)

Average and conditional effects as functions of model parameters:

g1(z) = (β10 −β00)+ (β11 −β01) ·z =γ10 +γ11 ·z

E[g1(Z) |X = 1] = (β10 −β00)+ (β11 −β01) ·µ11

E(Z) =µ01 ·P(X = 0)+µ11 ·P(X = 1)

E[g1(Z)] = (β10 −β00)+ (β11 −β01) ·E(Z)

Example 1

Mplus syntax for example 1:

TITLE:
Example 01 with continuous covariate

DATA:
file is data01.dat;
type is individual;

VARIABLE:
names are y x z;
classes = cell(2);
knownclass = cell(x=0 x=1);

ANALYSIS:
type = mixture;

Example 1

Mplus syntax for example 1 (continued):

MODEL:

%overall%
y on z; ! regression of y on z
[cell#1] (lc1); ! coefficient for group size 0

%cell#1%
y ON z* (b10); ! regression y on z in group x=0
[y*] (b00); ! name for intercept of this regression
[z*] (m0); ! conditional mean of z in group x=0

%cell#2%
y ON z* (b11); ! regression y on z in group x=1
[y*] (b01); ! name for intercept of this regression
[z*] (m1); ! conditional mean of z in group x=1

Example 1

Mplus syntax for example 1 (continued):

MODEL CONSTRAINT:
new(ga10 ga11); ! coefficients of effect function
new(px0 px1 mz); ! P(X=0), P(X=1) and unconditional mean of z
new(ae aet aeut ep1sd em1sd); ! various effects (see below)

ga10 = b01-b00;
ga11 = b11-b10;

px0 = exp(lc1)/(1+exp(lc1));
px1 = 1-px0;
mz = m0*px0 + m1*px1;

ae = ga10 + ga11*mz; ! average effect
aet = ga10 + ga11*m1; ! average effect on the treated
aeut = ga10 + ga11*m0; ! average effect on the untreated
ep1sd = ga10 + ga11*(mz + 2); ! effect given mean z + 1 SD
em1sd = ga10 + ga11*(mz - 2); ! effect given mean z - 1 SD

Example 2

Example 2

Example 2

Generalized ANCOVA with three treatment conditions and a single categorical covariate K
with three values:

E(Y |X ,K ) = g0(K )+g1(K ) · IX=1 +g2(K ) · IX=2

g0(K ) =γ000 +γ010 · IK=1 +γ020 · IK=2

g1(K ) =γ100 +γ110 · IK=1 +γ120 · IK=2

g2(K ) =γ200 +γ210 · IK=1 +γ220 · IK=2.

Some average and conditional effects:

E[g1(K )] =γ100+γ110·P(K = 1)+γ120·P(K = 2) Average effect 1 vs. 0
E[g2(K )] =γ200+γ210·P(K = 1)+γ220·P(K = 2) Average effect 2 vs. 0
g1(0) =γ100 Conditional effect 1 vs. 0 given K = 0
g2(2) =γ200 +γ220 Conditional effect 2 vs. 0 given K = 2
E[g1(K ) |X = 1] =γ100+γ110 ·P(K = 1 |X = 1)+
γ120 ·P(K = 2 |X = 1)

Effect 1 vs. 0 on the treated

Example 2

Multigroup SEM:(
Y

)= (
αxk0

)+ (
ζ1

)
Structural model for each cell C = c

P(C = m) = exp(λm)

1+
8∑

i=1
exp(λi)

m = 1,2, . . . ,8 Group sizes

P(C = 9) = 1−
8∑

m=1
P(C = m) Group size reference group

Example 2

In an intermediate step, we compute the regression coefficients of
EX=x(Y |K ) =βx00 +βx10 · IK=1 +βx20 · IK=2 and then the coefficients for the effect functions
(see below).

Computations for some average and conditional effects as functions of model parameters
(only shown for g1(K ), but similar for g2(K ) – see Mplus syntax for details):

γ100 =β100 −β000 =α100 −α000

γ110 =β110 −β010 = (α110 −α100)− (α010 −α000)

γ120 =β120 −β020 = (α120 −α100)− (α020 −α000)

P(K = 1) = P(X = 0,K = 1)+P(X = 1,K = 1)+P(X = 2,K = 1)

P(K = 2) = P(X = 0,K = 2)+P(X = 1,K = 2)+P(X = 2,K = 2)

E[g1(K )] =γ100 +γ110 ·P(K = 1)+γ120 ·P(K = 2)

g1(0) =γ100

Example 2

Mplus syntax for example 2:

TITLE: Example 02: nonortho treat and neediness
DATA: FILE IS data02.dat;
DEFINE:
! compute knowclass variable C
IF K == 0 AND X == 0 THEN C = 1 ;
IF K == 1 AND X == 0 THEN C = 2 ;
IF K == 2 AND X == 0 THEN C = 3 ;
IF K == 0 AND X == 1 THEN C = 4 ;
IF K == 1 AND X == 1 THEN C = 5 ;
IF K == 2 AND X == 1 THEN C = 6 ;
IF K == 0 AND X == 2 THEN C = 7 ;
IF K == 1 AND X == 2 THEN C = 8 ;
IF K == 2 AND X == 2 THEN C = 9 ;

VARIABLE: NAMES = Y X K ; ! variable names
USEVARIABLES = Y C;

CLASSES = cell( 9 ); ! 9 cells (3x3 combinations of X and K)
KNOWNCLASS = cell( C=1 C=2 C=3 C=4 C=5 C=6 C=7 C=8 C=9 );

ANALYSIS: TYPE = MIXTURE;

Example 2

Mplus syntax for example 2 (continued):

MODEL:

%OVERALL%
Y; ! We just estimate mean and variance of Y in each cell

! coefficients for group sizes 1-8 (logistic metric)
! group 9 is the reference group
[cell#1] (LC1);
[cell#2] (LC2);
[cell#3] (LC3);
[cell#4] (LC4);
[cell#5] (LC5);
[cell#6] (LC6);
[cell#7] (LC7);
[cell#8] (LC8);

Example 2

Mplus syntax for example 2 (continued):

! mean of Y in each group
%cell#1%
[Y] (a0_0_0); ! intercept/mean of Y
%cell#2%
[Y] (a0_1_0);
%cell#3%
[Y] (a0_2_0);
%cell#4%
[Y] (a1_0_0);
%cell#5%
[Y] (a1_1_0);
%cell#6%
[Y] (a1_2_0);
%cell#7%
[Y] (a2_0_0);
%cell#8%
[Y] (a2_1_0);
%cell#9%
[Y] (a2_2_0);

Example 2

Mplus syntax for example 2 (continued):

MODEL CONSTRAINT:

NEW(PC1 PC2 PC3 PC4 PC5 PC6 PC7 PC8 PC9 ! P(C=c)
PK1 PK2 ! P(K=k)
b0_0_0 b0_1_0 b0_2_0
b1_0_0 b1_1_0 b1_2_0
b2_0_0 b2_1_0 b2_2_0
g1_0_0 g1_1_0 g1_2_0 ! coefficients of effect function g1
g2_0_0 g2_1_0 g2_2_0 ! coefficients of effect function g2
Eg1 Eg2 ! average effect 1 vs 0 and average effect 2 vs. 0
Eg1gK0 Eg1gK1 Eg1gK2 ! effect 1 vs 0 given values k of K
Eg2gK0 Eg2gK1 Eg2gK2 ! effect 2 vs 0 given values k of K
);

Example 2

Mplus syntax for example 2 (continued):

! P(C=c)
PC1 = EXP(LC1) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +EXP(LC6)
+EXP(LC7) +EXP(LC8) +1);
PC2 = EXP(LC2) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +EXP(LC6)
+EXP(LC7) +EXP(LC8) +1);
PC3 = EXP(LC3) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +EXP(LC6)
+EXP(LC7) +EXP(LC8) +1);
PC4 = EXP(LC4) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +EXP(LC6)
+EXP(LC7) +EXP(LC8) +1);
PC5 = EXP(LC5) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +EXP(LC6)
+EXP(LC7) +EXP(LC8) +1);
PC6 = EXP(LC6) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +EXP(LC6)
+EXP(LC7) +EXP(LC8) +1);
PC7 = EXP(LC7) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +EXP(LC6)
+EXP(LC7) +EXP(LC8) +1);
PC8 = EXP(LC8) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +EXP(LC6)
+EXP(LC7) +EXP(LC8) +1);
PC9 = 1 - (PC1 + PC2 + PC3 + PC4 + PC5 + PC6 + PC7 + PC8);

Example 2

Mplus syntax for example 2 (continued):

! coefficients of (X=x)-conditional regressions of Y on K
! e.g. E^{X=0}(Y|K) = b0_0_0 + b0_1_0 * I_{K=1} + b0_2_0 * I_{K=2}
b0_0_0 = a0_0_0;
b0_1_0 = a0_1_0 - a0_0_0;
b0_2_0 = a0_2_0 - a0_0_0;
b1_0_0 = a1_0_0;
b1_1_0 = a1_1_0 - a1_0_0;
b1_2_0 = a1_2_0 - a1_0_0;
b2_0_0 = a2_0_0;
b2_1_0 = a2_1_0 - a2_0_0;
b2_2_0 = a2_2_0 - a2_0_0;

! coefficients of effect functions g1 and g2
g1_0_0 = b1_0_0 - b0_0_0;
g1_1_0 = b1_1_0 - b0_1_0;
g1_2_0 = b1_2_0 - b0_2_0;
g2_0_0 = b2_0_0 - b0_0_0;
g2_1_0 = b2_1_0 - b0_1_0;
g2_2_0 = b2_2_0 - b0_2_0;

Example 2

Mplus syntax for example 2 (continued):

! P(K=k)
PK1 = PC2 + PC5 + PC8;
PK2 = PC3 + PC6 + PC9;

! effects
Eg1 = g1_0_0 + g1_1_0*PK1 + g1_2_0*PK2; ! average effect 1 vs. 0
Eg2 = g2_0_0 + g2_1_0*PK1 + g2_2_0*PK2; ! average effect 2 vs. 0

Eg1gK0 = g1_0_0; ! effect 1 vs. 0 given K=0
Eg1gK1 = g1_0_0 + g1_1_0; ! effect 1 vs. 0 given K=1
Eg1gK2 = g1_0_0 + g1_2_0; ! effect 1 vs. 0 given K=2

Eg2gK0 = g2_0_0; ! effect 2 vs. 0 given K=0
Eg2gK1 = g2_0_0 + g2_1_0; ! effect 2 vs. 0 given K=1
Eg2gK2 = g2_0_0 + g2_2_0; ! effect 2 vs. 0 given K=2

Example 3

Example 3

Example 3:

η2: latent outcome variable with three indicators Y12,Y22,Y32

X : treatment variable (control, treat1, treat2)

η1: latent continuous covariate (pre-test) with three indicators Y11,Y21,Y31

K : gender (male, female)

E(η2 |X ,K ,η1) = g0(K ,η1)+g1(K ,η1) · IX=1 +g2(K ,η1) · IX=2

g0(K ,η1) =γ000 +γ010 ·K +γ001 ·η1 +γ011 ·K ·η1

g1(K ,η1) =γ100 +γ110 ·K +γ101 ·η1 +γ111 ·K ·η1

g2(K ,η1) =γ200 +γ210 ·K +γ201 ·η1 +γ211 ·K ·η1

Example 3

Effect functions (repeated):

g1(K ,η1) =γ100 +γ110 ·K +γ101 ·η1 +γ111 ·K ·η1

g2(K ,η1) =γ200 +γ210 ·K +γ201 ·η1 +γ211 ·K ·η1

Some average and conditional effects:

Average effect of X = 1 vs. X = 0:

E[g1(K ,η1)] =γ100 +γ110 ·E(K )+γ101 ·E(η1)+γ111 ·E(K ·η1)

Average effect of X = 2 vs. X = 0:

E[g2(K ,η1)] =γ200 +γ210 ·E(K )+γ201 ·E(η1)+γ211 ·E(K ·η1)

Conditional effect of X = 1 vs. X = 0 given K = 0:

E[g1(K ,η1) |K = 0] =γ100 +γ101 ·E(η1 |K = 0)

Conditional effect of X = 1 vs. X = 0 given X = 2:

E[g1(K ,η1) |X = 2] =γ100 +γ110 ·E(K |X = 2)+γ101 ·E(η1 |X = 2)+γ111 ·E(K ·η1 |X = 2)

Example 3

Multigroup SEM: Group-invariant τ-equivalent measurement model

Y11
Y21
Y31
Y12
Y22
Y32

=



0
0
0
0
0
0

+



1 0
1 0
1 0
0 1
0 1
0 1


(
η1
η2

)
+



ε11
ε21
ε31
ε12
ε22
ε32


Multigroup SEM: Group-specific structural model in each of the six groups (2 x 3)(

η2
η1

)
=

(
αxk0
µxk1

)
+

(
αxk1

0

)
η1 +

(
ζ1
ζ2

)
Multigroup SEM: Group weights for each of the six groups (2 x 3)

P(C = m) = exp(λm)

1+
5∑

i=1
exp(λi)

m = 1,2, . . . ,5 Group sizes

P(C = 6) = 1−
5∑

m=1
P(C = m) Group size reference group

Example 3

Based on the parameters of the model (previous slide), we can estimate average and
conditional effects. I am not showing all details here but see Mplus syntax.

Some examples:

γ100 =α100 −α000

γ101 =α101 −α001

γ110 = (α110 −α100)− (α010 −α000)

γ111 = (α111 −α101)− (α011 −α000)

E(η1) =µ001 ·P(X = 0,K = 0)+µ011 ·P(X = 0,K = 1)+ . . .+µ211 ·P(X = 2,K = 1)

P(K = 1) = P(X = 0,K = 1)+P(X = 1,K = 1)+P(X = 2,K = 1)

P(X = 0) = P(X = 0,K = 0)+P(X = 0,K = 1)+P(X = 0,K = 2)

Example 3

Mplus syntax for example 3:

TITLE: Example 03 with latent variables and categorical covariate
DATA: FILE IS data03.dat;
DEFINE:
! compute knowclass variable C whose values represent all
! combinations of values of X and K
IF K == 0 AND X == 0 THEN C = 1 ;
IF K == 1 AND X == 0 THEN C = 2 ;
IF K == 0 AND X == 1 THEN C = 3 ;
IF K == 1 AND X == 1 THEN C = 4 ;
IF K == 0 AND X == 2 THEN C = 5 ;
IF K == 1 AND X == 2 THEN C = 6 ;

VARIABLE: NAMES = K X y32 y22 y12 eta2 y31 y21 y11 eta1 Ix2 Ix1;
USEVARIABLES = y32 y22 y12 y31 y21 y11 C;
MISSING =.;
CLASSES = cell( 6 ); ! 6 cells (3x2 combinations of X and K)
KNOWNCLASS = cell( C=1 C=2 C=3 C=4 C=5 C=6 );

ANALYSIS: TYPE = MIXTURE;

Example 3

Mplus syntax for example 3 (continued):

MODEL:
%OVERALL%
! measurement model for latent variables
eta2 by y12@1 y22@1 y32@1;
eta1 by y11@1 y21@1 y31@1;
[y11@0 y21@0 y31@0 y12@0 y22@0 y32@0];
[eta2* eta1*];

! regression of eta2 on eta1
eta2 ON eta1;

! coefficients for group sizes 1-5 (logistic metric)
! group 6 is the reference group
[cell#1] (LC1);
[cell#2] (LC2);
[cell#3] (LC3);
[cell#4] (LC4);
[cell#5] (LC5);

Example 3

Mplus syntax for example 3 (continued):

! regression of eta2 on eta1 in each cell
%cell#1%
eta2 ON eta1(a0_0_1); ! regression coefficient
[eta1] (EzC1); ! cell specific mean of eta1
[eta2] (a0_0_0); ! intercept of regression

%cell#2%
eta2 ON eta1(a0_1_1);
[eta1] (EzC2);
[eta2] (a0_1_0);

%cell#3%
eta2 ON eta1(a1_0_1);
[eta1] (EzC3);
[eta2] (a1_0_0);

Example 3

Mplus syntax for example 3 (continued):

%cell#4%
eta2 ON eta1(a1_1_1);
[eta1] (EzC4);
[eta2] (a1_1_0);

%cell#5%
eta2 ON eta1(a2_0_1);
[eta1] (EzC5);
[eta2] (a2_0_0);

%cell#6%
eta2 ON eta1(a2_1_1);
[eta1] (EzC6);
[eta2] (a2_1_0);

Example 3

Mplus syntax for example 3 (continued):

MODEL CONSTRAINT:

NEW(PC1 PC2 PC3 PC4 PC5 PC6 PK1 ! P(C=c) and P(K=1)
Ez ! unconditional mean eta1
EzK1 ! mean of K*eta1 interaction term
b0_0_0 b0_0_1 b0_1_0 b0_1_1
b1_0_0 b1_0_1 b1_1_0 b1_1_1
b2_0_0 b2_0_1 b2_1_0 b2_1_1
g1_0_0 g1_0_1 g1_1_0 g1_1_1
g2_0_0 g2_0_1 g2_1_0 g2_1_1
Eg1 Eg2);

! cell probabilities
PC1 = EXP(LC1) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +1);
PC2 = EXP(LC2) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +1);
PC3 = EXP(LC3) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +1);
PC4 = EXP(LC4) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +1);
PC5 = EXP(LC5) / (EXP(LC1) +EXP(LC2) +EXP(LC3) +EXP(LC4) +EXP(LC5) +1);
PC6 = 1 - (PC1 +PC2 +PC3 +PC4 +PC5);

Example 3

Mplus syntax for example 3 (continued):

PK1 = PC2 +PC4 +PC6;
Ez = EzC1*PC1 +EzC2*PC2 +EzC3*PC3 +EzC4*PC4 +EzC5*PC5 +EzC6*PC6;
EzK1 = EzC2*PC2 + EzC4*PC4 + EzC6*PC6; ! mean of K*eta1 interaction term

! coefficients of (X=x)-conditional regressions of eta2 on K and eta1
b0_0_0 = a0_0_0;
b0_0_1 = a0_0_1;
b0_1_0 = a0_1_0 - a0_0_0;
b0_1_1 = a0_1_1 - a0_0_1;
b1_0_0 = a1_0_0;
b1_0_1 = a1_0_1;
b1_1_0 = a1_1_0 - a1_0_0;
b1_1_1 = a1_1_1 - a1_0_1;
b2_0_0 = a2_0_0;
b2_0_1 = a2_0_1;
b2_1_0 = a2_1_0 - a2_0_0;
b2_1_1 = a2_1_1 - a2_0_1;

Example 3

Mplus syntax for example 3 (continued):

! coefficients of effect functions g1 and g2
g1_0_0 = b1_0_0 - b0_0_0;
g1_0_1 = b1_0_1 - b0_0_1;
g1_1_0 = b1_1_0 - b0_1_0;
g1_1_1 = b1_1_1 - b0_1_1;

g2_0_0 = b2_0_0 - b0_0_0;
g2_0_1 = b2_0_1 - b0_0_1;
g2_1_0 = b2_1_0 - b0_1_0;
g2_1_1 = b2_1_1 - b0_1_1;

! average effects
Eg1 = g1_0_0 + g1_0_1*Ez + g1_1_0*PK1 + g1_1_1*EzK1;
Eg2 = g2_0_0 + g2_0_1*Ez + g2_1_0*PK1 + g2_1_1*EzK1;

Software

Software

Mplus can be used for the models presented here.

Since specification of the MODEL CONSTRAINT can be tedious for complex models, we
are developing an R package that does this automatically for you. The R package has not
been published yet, but can be obtained by emailing Lisa
(Lisa.Dietzfelbinger@uni-jena.de) or me (Axel.Mayer@ugent.be).

For those of you who also use other software, we developed a lavaan based version of
the program, which includes a graphical user interface and can be obtained via my
Github repository (https://github.com/amayer2010/effectliter).

Thank you for your attention.

